

**Les chaînes d'au moins 6 résultats consécutifs égaux<sup>1</sup> sur une série de 200 Piles-Faces  
SIMULEZ, Y' A TOUT À VOIR !<sup>2</sup>  
Un résultat probabiliste défiant l'intuition**

## Préambule

Lorsque j'ai découvert ce fameux résultat<sup>3</sup>, début des années 2000, j'ai vu en lui tout l'intérêt de la simulation... simulation permettant d'accéder à moindre coût à un résultat (estimation d'une probabilité en l'occurrence), non trivial (voir démonstration plus bas), et pour le moins assez surprenant...

À la croisée de plusieurs chemins (simulation, probabilités, tableur, algorithmique, suites, chaînes de Markov...), j'ai trouvé intéressant de faire ici un point, même si beaucoup de choses ont déjà été écrites sur le sujet (il suffit d'aller voir sur Internet !).

Par ailleurs, j'ai la faiblesse de penser, je le constate encore régulièrement, qu'encore beaucoup de collègues ne connaissent pas ce résultat surprenant.

Ma modeste contribution dans ce petit bilan, outre de rassembler des infos déjà existantes et d'en proposer une certaine organisation, aura été de proposer, avec le logiciel LARP, un traitement algorithmique, riche par la réflexion qu'il suscite pour sa mise en œuvre, et pertinent par la praticité et l'efficacité de son utilisation.

## 1. Le problème

Pour le prendre à l'envers (le problème), chaque année je lance en *Première S* le défi suivant : « *À la maison, chacun d'entre vous va jouer à Pile ou Face 200 fois, et noter sa liste des 200 résultats. Vous faites tous le jeu comme il faut, avec une pièce, SAUF l'un d'entre vous (« le tricheur », que vous choisirez sans me le dire), qui notera la série de 200 résultats qu'il veut, sans faire réellement les lancers avec la pièce. Et bien, je vous parie ce que vous voulez que je découvre le tricheur de la classe ! Ah, ah... ! À la prochaine... »* ».

À ce moment-là de l'histoire, aucun élève ne se doute du résultat contre-intuitif suivant : Sauf le tricheur qui n'osera pas écrire des séquences trop longues de résultats consécutifs égaux, pensant que le hasard fait bien les choses, qu'un certain équilibre doit être respecté au cours du temps, que la pièce a de la mémoire, etc.<sup>4</sup>, tous (avec un tout petit peu de chance<sup>5</sup> pour moi) les vrais joueurs auront dans leur série de 200 lancers, au moins une séquence d'au moins 6 résultats consécutifs égaux !

Je repèrerai alors facilement le tricheur : « Cette année, c'était Billy !? », « Gagné Monsieur !... ? Mais comment vous faites ? »

Voici donc le problème à l'endroit : « Sur une série de 200 lancers (Hypothèse de modèle = équiprobabilité sur Pi-Fa), évaluer la probabilité de l'événement "**Obtenir au moins une séquence d'au moins 6 résultats consécutifs égaux**" ».

<sup>1</sup> On parle parfois d'Idem-Séquence.

<sup>2</sup> Petit hommage à Michel Colucci, dit COLUCHE.

<sup>3</sup> Résultat ultra-connu, découvert, pour ma part, en lisant une brochure de référence : n°112 de la CII Lycée Technologique « *Enseigner la Statistique au lycée : des enjeux aux méthodes* », de J-L. PIEDNOIR et P. DUTARTE, 2001.

<sup>4</sup> Qui sont tant de conceptions erronées sur « hasard et probabilités ».

<sup>5</sup> La probabilité en question est de l'ordre de 0,965 !


## 2. Une simulation sur tableur<sup>6</sup>

Voici une simulation qui peut paraître simple une fois faite, mais qui est un exercice de recherche très intéressant, demandant quelque réflexion, à proposer aux les élèves...

On remarque, et il n'est jamais superflu de le souligner encore et toujours, que l'on simule toujours un modèle-source : ici on rentre avec **alea()** le modèle-source d'équiprobabilité sur Pi-Fa. L'objectif étant ensuite, "en fin de simulation", de récupérer la valeur (plus précisément, une estimation de la valeur) d'une probabilité, difficile d'accès sur le plan théorique.

On simule le lancer d'une pièce équilibrée (codé par 0/1)<sup>7</sup>

	A	B	C	D
1	simulation 1		simulation 2	
2	0	1	0	1
3	0	2	0	2
4	0	3	0	3




<sup>6</sup> On trouve, par exemple, cette simulation, dans l'excellent ouvrage de P. DUTARTE « *L'induction statistique au lycée, illustrée par le tableur* », chez Didier, 2005.

<sup>7</sup> Sur la plan pédagogique, certains, à raison, préféreront la formule  $SI(ALEA()<0,5;0;1)$  ; il peut être intéressant aussi de faire réfléchir sur celle écrite ici... et même sur celle-là d'ailleurs  $ALEA.ENTRE.BORNES(0;1)$ , la plus directe peut-être...


On crée le compteur du nombre de lancers consécutifs égaux :  
 On teste si un lancer est égal au précédent :  
 Si oui, on incrémente le compteur de 1  
 Sinon, on le réinitialise à 1

GR2 = =SI(GO2=GO1;GR1+1;1)




En bas de chaque série (colonne) de 200 lancers de pièces,  
 on relève "la hauteur maximale de vol" du compteur

A202 = =MAX(B2:B201)




Sur les 100 séries de 200 lancers, on compte celles pour lesquelles la « hauteur maximale de vol » du compteur a atteint ou dépassé 6.

A205 = =NB.SI(A202:GR202;">=6")/100 , au format % !




↓ Touche F9, qui recalculé la feuille de tableur, produisant ainsi une nouvelle simulation par modification des résultats des fonctions ALEA().



On obtient ainsi une estimation de la probabilité cherchée...  
 « Ça tourne autour de 0,95... »  
 On ne s'attendait pas à une valeur si élevée !  
 « Le hasard est une loi qui voyage incognito » (Proverbe arabe)

### 3. Une simulation par algorithme : traitement avec LARP

Module PRINCIPAL



**LongMax = Module auxiliaire** (sous-programme) qui, à une valeur entrante de la taille **n** de la série de lancers, renvoie la longueur **LMax** de l'Idem-Séquence de longueur maximale obtenue au cours de cette série.

## Pseudo-Code du module PRINCIPAL

DÉBUT

ÉCRIRE "Fournit, pour n lancers Pi-Fa simulés, une estimation de la probabilité d'obtenir au moins une chaîne d'au moins long résultats consécutifs égaux :"

ÉCRIRE ""

ÉCRIRE "Modifier, si besoin, les valeurs de n, long et NbSim dans le programme !"

ÉCRIRE "(Par défaut, n=200, long=6 et NbSim=2000)"

n=200

long=6

NbSim=2000

CS=0

POUR j = 1 JUSQU'À NbSim INCRÉMENT 1 FAIRE

    LMax=EXÉCUTER LongMax(n)

    SI LMax>=long ALORS

        CS=CS+1

    FINSI

FINPOUR

ÉCRIRE ""

ÉCRIRE "Une estimation de la probabilité d'obtenir, sur",n,"lancers de Pi-Fa, au moins une chaîne d'au moins",long,"résultats consécutifs égaux, est : ",CS/NbSim

FIN

## Description du module PRINCIPAL

Il s'agit donc de faire **NbSim** (2 000 par défaut) simulations de **n** (200 par défaut) lancers d'une pièce équilibrée : les NbSim simulations se font avec la boucle **POUR** et les n lancers se font dans le *module auxiliaire* qui renvoie en sortie **LMax**.

Un compteur est installé pour compter, parmi les NbSim simulations, le nombre de Succès = Lmax (longueur de l'Idem-Séquence de longueur maximale sur une série de taille n)  $\geq 6$ .

On peut donc estimer alors la probabilité cherchée par la fréquence de ce succès (Loi des grands nombres).


## Description du module auxiliaire

On jette une première fois la pièce : **Plancher(alea+0.5)** est à LARP ce que Ent(Alea()+0,5) est au tableur ; sachant qu'on peut aussi directement écrire **alea(0,1)** qui correspond lui à alea.entre.bornes(0;1).

On effectue les n-1 autres lancers à l'aide d'une boucle **POUR**.

On crée alors un compteur qui va évaluer la longueur de chaque Idem-Séquence, et qu'on initialise donc à 1 en dehors de la boucle POUR. Ce compteur subit une incrémentation de 1 **SI** un Jet aboutit à la même issue que le Jet précédent, mais est aussi de suite réinitialisé à 1 **SINON** (c'est-à-dire dès qu'une alternance se produit sur les issues).

**LMax = MAX(Compteur,LMax)**, initialisé à 1 en dehors de la boucle POUR, va donc toujours garder en mémoire la longueur de l'Idem-Séquence de longueur maximale obtenue "jusque-là".




Pseudo-code de LongMax

```

ENTRER n
Jet1=Plancher(alea+0.5)
Compteur=1
LMax=1
POUR i = 1 JUSQU'À n-1 INCRÉMENT 1 FAIRE
  Jet2=Plancher(alea+0.5)
  SI Jet2=Jet1 ALORS
    Compteur=Compteur+1
    LMax=MAX(Compteur, LMax)
  SINON
    Compteur=1
  FINSI
  Jet1=Jet2
FINPOUR
RETOURNER LMax
  
```


Copier au presse-papiers    ? Aide    Fermer

### Le résultat après une exécution...




```
Console
Terminé
Fournit, pour n lancers Pi-Fa simulés, une estimation de la probabilité d'obtenir au moins une chaîne d'au moins long résultats consécutifs égaux :
Modifier, si besoin, les valeurs de n, long et NbSim dans le programme !
(Par défaut, n=200, long=6 et NbSim=2000)
Une estimation de la probabilité d'obtenir, sur 200 lancers de Pi-Fa, au moins une chaîne d'au moins 6 résultats consécutifs égaux, est : 0.9655
Appuyez sur une touche pour fermer la console..._
```

### Le résultat après une autre exécution...



```
Console
Terminé
Fournit, pour n lancers Pi-Fa simulés, une estimation de la probabilité d'obtenir au moins une chaîne d'au moins long résultats consécutifs égaux :
Modifier, si besoin, les valeurs de n, long et NbSim dans le programme !
(Par défaut, n=200, long=6 et NbSim=2000)
Une estimation de la probabilité d'obtenir, sur 200 lancers de Pi-Fa, au moins une chaîne d'au moins 6 résultats consécutifs égaux, est : 0.9665
Appuyez sur une touche pour fermer la console..._
```

### Une dernière pour la route...



```
Console
Terminé
Fournit, pour n lancers Pi-Fa simulés, une estimation de la probabilité d'obtenir au moins une chaîne d'au moins long résultats consécutifs égaux :
Modifier, si besoin, les valeurs de n, long et NbSim dans le programme !
(Par défaut, n=200, long=6 et NbSim=2000)
Une estimation de la probabilité d'obtenir, sur 200 lancers de Pi-Fa, au moins une chaîne d'au moins 6 résultats consécutifs égaux, est : 0.9705
Appuyez sur une touche pour fermer la console..._
```

#### 4. Une simulation par algorithme : traitement avec R<sup>8</sup> et représentations graphiques

---

R offre des possibilités de simulation, de traitement des séries simulées et de représentation graphique inégalées, propres à un logiciel de Statistique professionnelle.

Les algorithmes, programmes R et résultats correspondants figurent dans le document **ProblemesChainesL6.pdf**.

Ces programmes sont la mise en œuvre d'algorithmes fournissant :

- une estimation de la probabilité d'obtenir au moins une Idem-Séquence de longueur 6 ou plus ;
- la distribution simulée de la variable LMax (longueur de l'Idem-Séquence de longueur maximale) ;
- une illustration graphique des résultats des simulations qui représente les piles (en vert) ou faces (en bleu) obtenus, ainsi que les Idem-Séquences de longueur 6 ou plus (en rouge).

La recherche de l'algorithme permet de donner du sens à la notion d'Idem-Séquence et l'illustration graphique, de visualiser l'apparition de ces Idem-Séquences dans les simulations. Une réflexion sur les résultats de l'algorithme peut être menée en remarquant que dans la plupart des simulations, les Idem-Séquences rouges (de longueur supérieure ou égale à 6) sont entourées d'Idem-Séquences de même couleur, mais qu'il arrive que ce ne soit pas le cas. J'avais d'abord pensé à une erreur, mais il n'en est rien. On est juste dans une situation particulière que je laisse au lecteur le soin de découvrir...

La fonction particulière `rle(...)` compte toutes les Idem-Séquences dans une série statistique. Elle permet de simplifier l'algorithme et la programmation.

---

<sup>8</sup> Partie rédigée par Hubert RAYMONDAUD.

## 5. Les différents cadres de la démonstration<sup>9</sup>

### Problème


On lance 200 fois de suite une pièce de monnaie équilibrée et on s'intéresse à la probabilité d'obtenir au moins une idem-séquence de longueur 6 (suite de 6 PILE ou de 6 FACE consécutifs).

### Modélisation

Définissons les états dans lesquels peut se trouver une série de lancers de pièces.

Soit pour  $1 \leq i \leq 5$ ,  $E_i$  l'état "la suite de lancers se termine par une idem-séquence de longueur  $i$  et ne contient aucune idem-séquence de longueur 6" et soit F l'état "la suite de lancers contient au moins une idem-séquence de longueur 6".

Représentons l'évolution des lancers par un graphe probabiliste : à chaque lancer, on se trouve dans un des états  $E_i$  ou F avec des probabilités de transition indiquées sur le graphe suivant :



Examinons la situation après  $n$  lancer(s) ( $1 \leq n \leq 200$ ).

Notons  $p_n^i$  la probabilité que la série des  $n$  lancers se trouve dans l'état  $E_i$  et  $p_n$  la probabilité qu'elle soit dans l'état F.

L'état initial est donné par  $(p_1^1, p_1^2, p_1^3, p_1^4, p_1^5, p_1^F) = (1, 0, 0, 0, 0, 0)$ .

$$\text{On pose } X_n = (p_n^1, p_n^2, p_n^3, p_n^4, p_n^5, p_n^F), \text{ pour } 1 \leq n \leq 200, \text{ et } T = \begin{pmatrix} \frac{1}{2} & \frac{1}{2} & 0 & 0 & 0 & 0 \\ \frac{1}{2} & 0 & \frac{1}{2} & 0 & 0 & 0 \\ \frac{1}{2} & 0 & 0 & \frac{1}{2} & 0 & 0 \\ \frac{1}{2} & 0 & 0 & 0 & \frac{1}{2} & 0 \\ \frac{1}{2} & 0 & 0 & 0 & 0 & \frac{1}{2} \\ 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

la matrice de transition du système. On a la relation de récurrence  $X_{n+1} = X_n T$  pour  $1 \leq n \leq 199$ .

Ainsi l'état du système après 200 lancers est modélisé par  $X_{200} = X_1 T^{199}$ .

<sup>9</sup> Partie rédigée par Brigitte CHAPUT, Formatrice en mathématiques à l'ENFA de Toulouse.

## Réponse avec un logiciel de calcul formel

Un logiciel de calcul formel (ici DERIVE6) donne :

$$p_{200}^1 = \frac{14\ 176\ 410\ 776\ 452\ 266\ 653\ 026\ 399\ 153\ 191\ 100\ 171\ 036\ 247\ 629\ 812\ 728\ 621\ 565}{803\ 469\ 022\ 129\ 495\ 137\ 770\ 981\ 046\ 170\ 581\ 301\ 261\ 101\ 496\ 891\ 396\ 417\ 650\ 688}$$

$$p_{200}^2 = \frac{7\ 210\ 978\ 657\ 628\ 137\ 907\ 257\ 624\ 544\ 745\ 936\ 474\ 597\ 891\ 413\ 647\ 132\ 893\ 983}{803\ 469\ 022\ 129\ 495\ 137\ 770\ 981\ 046\ 170\ 581\ 301\ 261\ 101\ 496\ 891\ 396\ 417\ 650\ 688}$$

$$p_{200}^3 = \frac{229\ 246\ 201\ 756\ 953\ 867\ 916\ 083\ 703\ 292\ 740\ 244\ 901\ 382\ 835\ 947\ 994\ 921\ 817}{50\ 216\ 813\ 883\ 093\ 446\ 110\ 686\ 315\ 385\ 661\ 331\ 328\ 818\ 843\ 555\ 712\ 276\ 103\ 168}$$

$$p_{200}^4 = \frac{233\ 216\ 925\ 536\ 268\ 102\ 439\ 493\ 159\ 625\ 656\ 732\ 729\ 867\ 457\ 317\ 668\ 311\ 713}{100\ 433\ 627\ 766\ 186\ 892\ 221\ 372\ 630\ 771\ 322\ 662\ 657\ 637\ 687\ 111\ 424\ 552\ 206\ 336}$$

$$p_{200}^5 = \frac{237\ 256\ 425\ 361\ 644\ 497\ 110\ 729\ 629\ 252\ 980\ 854\ 088\ 616\ 335\ 025\ 871\ 956\ 465}{200\ 867\ 255\ 532\ 373\ 784\ 442\ 745\ 261\ 542\ 645\ 325\ 315\ 275\ 374\ 222\ 849\ 104\ 412\ 672}$$

$$p_{200}^6 = \frac{96\ 949\ 866\ 545\ 195\ 843\ 564\ 510\ 102\ 428\ 242\ 905\ 427\ 356\ 478\ 434\ 265\ 475\ 383\ 313}{100\ 433\ 627\ 766\ 186\ 892\ 221\ 372\ 630\ 771\ 322\ 662\ 657\ 637\ 687\ 111\ 424\ 552\ 206\ 336}$$

Une valeur approchée de la probabilité  $p_{200}$  cherchée est 0,965 312 801 1.

## Programmation de la suite des 6 probabilités sur tableur

On programme la suite récurrente: 
$$\left\{ \begin{array}{l} p_{n+1}^1 = \frac{1}{2} p_n^1 + \frac{1}{2} p_n^2 + \frac{1}{2} p_n^3 + \frac{1}{2} p_n^4 + \frac{1}{2} p_n^5 \\ p_{n+1}^2 = \frac{1}{2} p_n^1 \\ p_{n+1}^3 = \frac{1}{2} p_n^2 \\ p_{n+1}^4 = \frac{1}{2} p_n^3 \\ p_{n+1}^5 = \frac{1}{2} p_n^4 \\ p_{n+1}^6 = \frac{1}{2} p_n^5 \\ (p_1^1, p_1^2, p_1^3, p_1^4, p_1^5, p_1^6) = (1, 0, 0, 0, 0, 0) \end{array} \right. \quad \text{pour } n \in \mathbb{N}^*$$

avec les formules :

	A	B	C	D	E	F	G
1	n	$p_n^1$	$p_n^2$	$p_n^3$	$p_n^4$	$p_n^5$	$p_n^6$
2	1	1	0	0	0	0	0
3	2	=0,5*(B2+C2+D2+E2+F2)	=0,5*B2	=0,5*C2	=0,5*D2	=0,5*E2	=0,5*F2+G2
4	3	=0,5*(B3+C3+D3+E3+F3)	=0,5*B3	=0,5*C3	=0,5*D3	=0,5*E3	=0,5*F3+G3
5	4	=0,5*(B4+C4+D4+E4+F4)	=0,5*B4	=0,5*C4	=0,5*D4	=0,5*E4	=0,5*F4+G4
6	5	=0,5*(B5+C5+D5+E5+F5)	=0,5*B5	=0,5*C5	=0,5*D5	=0,5*E5	=0,5*F5+G5
7	6	=0,5*(B6+C6+D6+E6+F6)	=0,5*B6	=0,5*C6	=0,5*D6	=0,5*E6	=0,5*F6+G6
8	7	=0,5*(B7+C7+D7+E7+F7)	=0,5*B7	=0,5*C7	=0,5*D7	=0,5*E7	=0,5*F7+G7
9	8	=0,5*(B8+C8+D8+E8+F8)	=0,5*B8	=0,5*C8	=0,5*D8	=0,5*E8	=0,5*F8+G8
10	9	=0,5*(B9+C9+D9+E9+F9)	=0,5*B9	=0,5*C9	=0,5*D9	=0,5*E9	=0,5*F9+G9
11	10	=0,5*(B10+C10+D10+E10+F10)	=0,5*B10	=0,5*C10	=0,5*D10	=0,5*E10	=0,5*F10+G10
12	11	=0,5*(B11+C11+D11+E11+F11)	=0,5*B11	=0,5*C11	=0,5*D11	=0,5*E11	=0,5*F11+G11

pour obtenir :

G201		=0,5*F200+G200						
	A	B	C	D	E	F	G	H
1	$n$	$p_n^1$	$p_n^2$	$p_n^3$	$p_n^4$	$p_n^5$	$p_n$	
2	1	1	0	0	0	0	0	
3	2	0,5	0,5	0	0	0	0	
4	3	0,5	0,25	0,25	0	0	0	
5	4	0,5	0,25	0,125	0,125	0	0	
6	5	0,5	0,25	0,125	0,0625	0,0625	0	
7	6	0,5	0,25	0,125	0,0625	0,03125	0,03125	
8	7	0,484375	0,25	0,125	0,0625	0,03125	0,046875	
9	8	0,4765625	0,2421875	0,125	0,0625	0,03125	0,0625	
10	9	0,46875	0,23828125	0,12109375	0,0625	0,03125	0,078125	

et une valeur approchée de la probabilité cherchée dans la dernière cellule :

$$p_{200} \approx 0,965\ 312\ 8.$$

G201		=0,5*F200+G200						
	A	B	C	D	E	F	G	H
197	196	0,01889856	0,00961295	0,00488973	0,00248721	0,00126515	0,9628464	
198	197	0,0185768	0,00944928	0,00480648	0,00244486	0,00124361	0,96347897	
199	198	0,01826051	0,00928884	0,00472464	0,00240324	0,00122243	0,96410078	
200	199	0,01794961	0,00913026	0,0046442	0,00236232	0,00120162	0,96471199	
201	200	0,017644	0,00897481	0,00456513	0,0023221	0,00118116	0,9653128	
202								

Voir feuille *Formule récurrence* du classeur joint **IdemséquenceChaput.xls**.

### Calcul matriciel et tableur

On utilise la fonction PRODUITMAT d'un tableur qui permet d'effectuer des produits de matrice.

**Étape 1 :** On saisit la matrice  $T$  dans la plage B1:G6.

**Étape 2 :** On saisit les probabilités de l'état initial dans la plage B9:G9.

B10		={PRODUITMAT(B9:G9;\$B\$1:\$G\$6)}						
	A	B	C	D	E	F	G	H
1		0,5	0,5	0	0	0	0	
2		0,5	0	0,5	0	0	0	
3	Matrice T	0,5	0	0	0,5	0	0	
4		0,5	0	0	0	0,5	0	
5		0,5	0	0	0	0	0,5	
6		0	0	0	0	0	1	
7								
8	$n$	$p_n^1$	$p_n^2$	$p_n^3$	$p_n^4$	$p_n^5$	$p$	
9	1	1	0	0	0	0	0	Etat initial
10	2	0,5	0,5	0	0	0	0	
11	3	0,5	0,25	0,25	0	0	0	

**Étape 3 :** On calcule les probabilités de l'état 2 dans la plage B10:G10. Pour cela on sélectionne la plage B10:G10 et on saisit la formule =PRODUITMAT(B9:G9;\$B\$1:\$G\$6) que l'on valide par les trois touches simultanées **[Ctrl]+[MAJ]+[Entrée]**. Cela a pour effet de créer une formule matricielle (matérialisée par des accolades autour) qui rend solidaires les 6 cellules de la plage et qui leur affecte les termes de la matrice produit résultat. On peut ensuite copier la formule jusqu'à la plage B208:G208 et obtenir une valeur approchée de la probabilité cherchée dans la dernière cellule :  $p_{200} \approx 0,965\ 312\ 8$ .

	A	B	C	D	E	F	G	H
G208	={=PRODUITMAT(B207:G207,\$B\$1:\$G\$6)}							
200	192	0,02024233	0,01029647	0,00523741	0,00266406	0,0013551	0,96020463	
201	193	0,01989769	0,01012116	0,00514824	0,0026187	0,00133203	0,96088218	
202	194	0,01955891	0,00994884	0,00506058	0,00257412	0,00130935	0,96154819	
203	195	0,0192259	0,00977946	0,00497442	0,00253029	0,00128706	0,96220287	
204	196	0,01889856	0,00961295	0,00488973	0,00248721	0,00126515	0,9628464	
205	197	0,0185768	0,00944928	0,00480648	0,00244486	0,00124361	0,96347897	
206	198	0,01826051	0,0092884	0,00472464	0,00240324	0,00122243	0,96410078	
207	199	0,01794961	0,00913026	0,0046442	0,00236232	0,00120162	0,96471199	
208	200	0,017644	0,00897481	0,00456513	0,0023221	0,00118116	0,9653128	
209								

Voir feuille *Calcul matriciel* du classeur joint **IdemSéquenceChaput.xls**.